Patterns of recursion

Readings: none.

Topics:

- Simple recursion
- Using accumulative recursion
- Recognizing generative recursion
All of the recursion we have done to date has followed a pattern we call **simple recursion**.

The templates we have been using have been derived from a data definition and specify the form of the recursive application.

We will now learn to use a new pattern of recursion, **accumulative recursion**, and learn to recognize **generative recursion**.

For the next several lecture modules we will use simple recursion and accumulative recursion. We will avoid generative recursion until the end of the course.
Simple recursion

Recall from Module 06:

In **simple recursion**, every argument in a recursive function application (or applications, if there are more than one) are either:

- unchanged, or
- *one step* closer to a base case according to a data definition
The limits of simple recursion

;; (max-list-v2 lon) produces the maximum element of lon
;; Examples:
(check-expect (max-list-v2 '(6 2 3 7 1)) 7)

;; max-list-v2: (listof Num) → Num
;; Requires: lon is nonempty
(define (max-list-v2 lon)
  (cond [(empty? (rest lon)) (first lon)]
        [(> (first lon) (max-list-v2 (rest lon))) (first lon)]
        [else (max-list-v2 (rest lon))]))

There may be two recursive applications of max-list.
The code for `max-list-v2` is correct.

But computing `(max-list-v2 (countup-to 1 25))` is very slow.

Why?

The initial application is on a list of length 25.

There are two recursive applications on the rest of this list, which is of length 24.

Each of those makes two recursive applications.
max-list can make up to $2^n - 1$ recursive applications on a list of length $n$.

We informally call this **exponential blowup**.
Measuring efficiency

We can take the number of recursive applications as a rough measure of a function’s efficiency. \texttt{max-list-v2} can take up to \(2^n - 1\) recursive applications.

\texttt{length} makes \(n\) recursive applications on a list of length \(n\).

\texttt{length} is clearly more efficient than \texttt{max-list-v2}.

We say that \texttt{length}’s efficiency is proportional to \(n\) and \texttt{max-list-v2}’s efficiency is proportional to \(2^n\). We express the former as \(O(n)\) and the later as \(O(2^n)\).
There are “families” of algorithms with similar efficiencies. Examples, from most efficient to least:

<table>
<thead>
<tr>
<th>“Big-O”</th>
<th>Example</th>
</tr>
</thead>
<tbody>
<tr>
<td>$O(1)$</td>
<td>no recursive calls; tax-payable</td>
</tr>
<tr>
<td>$O(\lg n)$</td>
<td>divide in half, work on one half; binary-search on a balanced tree</td>
</tr>
<tr>
<td>$O(n)$</td>
<td>one recursive application for each item; length</td>
</tr>
<tr>
<td>$O(n \lg n)$</td>
<td>divide in half, work on both halves; quicksort</td>
</tr>
<tr>
<td>$O(n^2)$</td>
<td>an $O(n)$ application for each item; insertion-sort</td>
</tr>
<tr>
<td>$O(2^n)$</td>
<td>two recursive applications for each item; max-list</td>
</tr>
</tbody>
</table>

Much more about “Big-O” notation and efficiency in later courses.

$\lg$ is $\log_2$. 
Fast $O(n)$

(define (max-list-v1 lon)
  (cond [(empty? (rest lon)) (first lon)]
        [else (max (first lon) (max-list-v1 (rest lon))))])

Slow $O(2^n)$

(define (max-list-v2 lon)
  (cond [(empty? (rest lon)) (first lon)]
        [ (> (first lon) (max-list-v2 (rest lon))) (first lon)]
        [else (max-list-v2 (rest lon))])))
A human approach

Humans don’t seem to use either of the two versions of \texttt{max-list} shown earlier. Instead, we tend to find the maximum of a list of numbers by scanning it, remembering the largest value seen so far. When we see a value that’s larger than the largest seen so far, we remember the new value – until we see one that is still larger. When we get to the end of the list, the largest value seen so far is the largest value in the list.
Accumulative recursion

Computationally, we can pass down that largest value seen so far as a parameter called an *accumulator*.

This parameter accumulates the result of prior computation, and is used to compute the final answer that is produced in the base case.

This approach results in the code on the next slide.
(max-list/acc lon max-so-far) produces the largest of the maximum element of lon and max-so-far

(max-list/acc: (listof Num) Num → Num)

(define (max-list/acc lon max-so-far)
  (cond [(empty? lon) max-so-far]
        [(> (first lon) max-so-far)
          (max-list/acc (rest lon) (first lon))]
        [else (max-list/acc (rest lon) max-so-far)]))

(define (max-list-v3 lon)
  (max-list/acc (rest lon) (first lon)))
Now even \((\text{max-list2 (countup-to 1 200000)})\) is fast.

\[
\begin{align*}
(\text{max-list2 (list 1 2 3 9 5)}) \\
\Rightarrow (\text{max-list/acc (list 2 3 9 5) 1}) \\
\Rightarrow (\text{max-list/acc (list 3 9 5) 2}) \\
\Rightarrow (\text{max-list/acc (list 9 5) 3}) \\
\Rightarrow (\text{max-list/acc (list 5) 9}) \\
\Rightarrow (\text{max-list/acc (list ) 9}) \\
\Rightarrow 9
\end{align*}
\]
Accumulative recursion

This technique is known as **accumulative recursion**.

It is more difficult to develop and reason about such code, which is why simple recursion is preferable if it is appropriate.

HtDP discusses it much later than we are doing (after material we cover in lecture module 10) but in more depth.
> Indicators of the accumulative recursion pattern

- All arguments to recursive function applications are:
  - unchanged, or
  - one step closer to a base case in the data definition, or
  - a partial answer (passed in an *accumulator*).

- The value(s) in the accumulator(s) are used in one or more base cases.

- The accumulatively recursive function usually has a wrapper function that sets the initial value of the accumulator(s).
Using simple recursion:

```scheme
;;; my-reverse: (listof X) → (listof X)
(define (my-reverse lst)
  (cond
   [(empty? lst) empty]
   [else (append (my-reverse (rest lst))
                 (list (first lst)))]))
```

Intuitively, `append` does too much work in repeatedly moving over the produced list to add one element at the end.

This has the same worst-case behaviour as insertion sort, $O(n^2)$. 
Reversing a list with an accumulator

;;; (my-reverse lst) reverses lst using accumulative recursion
;;; Example:
(check-expect (my-reverse '(1 2 3)) '(3 2 1))

;;; my-reverse: (listof X) → (listof X)
(define (my-reverse lst) ; wrapper function
 (my-rev/acc lst empty))

(define (my-rev/acc lst acc) ; helper function
 (cond [(empty? lst) acc]
       [else (my-rev/acc (rest lst)
                        (cons (first lst) acc))]))

This is $O(n)$.  

Efficiency

17/23 09: Patterns
A condensed trace

```
> (my-reverse (list 1 2 3 4 5))
⇒ (my-rev/acc (list 1 2 3 4 5) empty)
⇒ (my-rev/acc (list 2 3 4 5) (cons 1 empty))
⇒ (my-rev/acc (list 3 4 5) (cons 2 (list 1)))
⇒ (my-rev/acc (list 4 5) (cons 3 (list 2 1)))
⇒ (my-rev/acc (list 5) (cons 4 (list 3 2 1)))
⇒ (my-rev/acc (list ) (cons 5 (list 4 3 2 1)))
⇒ (list 5 4 3 2 1)
```
Exercise 1

The $n$th Fibonacci number is the sum of the two previous Fibonacci numbers:

\[ f_0 = 0, \quad f_1 = 1, \quad f_n = f_{n-1} + f_{n-2}, \quad n \geq 2 \]

This can be implemented directly using simple recursion, as follows:

```scheme
(define (fib n)
  (cond [(< n 2) n]
        [else (+ (fib (- n 1)) (fib (- n 2)))]))
```

This works: \((\text{fib } 6) \Rightarrow 8, \ (\text{fib } 25) \Rightarrow 75025\). But \((\text{fib } 50)\) takes days!

Write a function \((\text{extend-fib } n \text{ lst})\) that consumes a \text{Nat} and a \text{(listof Nat)}. Given \text{lst}, a list containing at least 2 Fibonacci values in descending order, it returns a list containing \(n\) more Fibonacci values.

\((\text{extend-fib } 8 \text{ (list 1 0)}) \Rightarrow \text{(list 34 21 13 8 5 3 2 1 1 0)}\)
Exercise 2

Write a function \(\text{fiba } n\) that is a wrapper for \text{extend-fib}, and produces the \(n\)th Fibonacci number.

\[
\begin{align*}
(fiba \ 0) & \Rightarrow 0 \\
(fiba \ 50) & \Rightarrow 12586269025 \\
(fiba \ 250) & \Rightarrow 7896325826131730509282738943634332893686268675876375
\end{align*}
\]
Exercise 3

Given a list of Nums, use accumulative recursion to write $\text{mean}$, which produces the average (mean) of the list.

Hints:

- $\text{mean}$ will be a wrapper function.
- How many accumulators do you need?
Generative Recursion: GCD

In Math 135, you learn that the Euclidean algorithm for Greatest Common Divisor (GCD) can be derived from the following identity for $m > 0$:

$$gcd(n, m) = gcd(m, n \mod m)$$

We also have $gcd(n, 0) = n$.

We can turn this reasoning directly into a Racket function.
;; (euclid-gcd n m) computes gcd(n,m) using Euclidean algorithm

;; euclid-gcd: Nat Nat → Nat
(define (euclid-gcd n m)
  (cond [(zero? m) n]
        [else (euclid-gcd m (remainder n m))]))

This function does not use simple or accumulative recursion.
Generative recursion

The arguments in the recursive application were generated by doing a computation on m and n.

The function euclid-gcd uses generative recursion.

Once again, functions using generative recursion are easier to get wrong, harder to debug, and harder to reason about.

We will return to generative recursion in a later lecture module. Avoid generative recursion until then.
Simple vs. accumulative vs. generative recursion

In **simple recursion**, all arguments to the recursive function application (or applications, if there are more than one) are either unchanged, or *one step* closer to a base case in the data definition.

In **accumulative recursion**, parameters are as above, plus parameters containing partial answers used in the base case.

In **generative recursion**, parameters are freely calculated at each step. (Watch out for correctness and termination!)
Goals of this module

- You should be able to recognize uses of simple recursion, accumulative recursion, and generative recursion.
- You should be able to write functions using simple and accumulative recursion.
- You should know that some functions are much more efficient than others, that efficiency is expressed with “Big-O” notation, and that you’ll learn more about this in future courses.